# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Contributor: Kwpolska <kwpolska@kwpolska.tk>

pkgname=python-unidecode
pkgver=1.3.6
pkgrel=2
pkgdesc='ASCII transliterations of Unicode text'
arch=(any)
url=https://github.com/avian2/unidecode
license=(GPL)
depends=(python)
makedepends=(
  git
  python-setuptools
)
_tag=414199246e4871669896756279e3460ff0364839
source=(git+https://github.com/avian2/unidecode.git#tag=${_tag})
sha256sums=(SKIP)

pkgver() {
  cd unidecode

  git describe --tags | sed 's/^unidecode-//'
}

build() {
  cd unidecode

  python setup.py build
}

package() {
  cd unidecode

  python setup.py install --root="${pkgdir}" --optimize=1 --skip-build
}

# vim: ts=2 sw=2 et:
